

Low-level functional GPU programming for parallel algorithms

Martin Dybdal Martin Elsman

University of Copenhagen
dybber@dybber.dk, mael@diku.dk

Bo Joel Svensson Mary Sheeran

Chalmers University of Technology
joels@chalmers.se, mary.sheeran@chalmers.se

Abstract

We present a Functional Compute Language (FCL) for low-level GPU programming. FCL is functional in style, which allows for easy composition of program fragments and thus easy prototyping and a high degree of code reuse. In contrast with projects such as Futhark, Accelerate, Harlan, Nessie and Delite, the intention is not to develop a language providing fully automatic optimizations, but instead to provide a platform that supports absolute control of the GPU computation and memory hierarchies. The developer is thus required to have an intimate knowledge of the target platform, as is also required when using CUDA/OpenCL directly.

FCL is heavily inspired by Obsidian. However, instead of relying on a multi-staged meta-programming approach for kernel generation using Haskell as meta-language, FCL is completely self-contained, and we intend it to be suitable as an intermediate language for data-parallel languages, including data-parallel parts of high-level array languages, such as R, Matlab, and APL.

We present a dynamic semantics suitable for understanding the performance characteristics of both FCL and Obsidian-style programs. Our aim is that FCL will be useful as a platform for developing new parallel algorithms, as well as a code-generation target-language.

1. Introduction

In recent years, several languages for general purpose, data-parallel computation on GPUs have been suggested [2, 4, 5, 10, 11, 14]. Most of these language developments have focused on providing users with high-level specifications of programs and performing a range of automatic optimisations. Often no cost-model is specified, and the language is thus a black box for users who want to reason about the performance of their programs. Parallel algorithms researchers are sidelined, as it is hard to reason about the actual efficiency and performance characteristics of algorithms. The user is decoupled from the hardware model, and cannot be sure whether an operation will result in a memory transaction or not. This makes unexpected performance hits hard to debug. Also, some algorithms require memory patterns not supported by the prevalent set of primitives, or depend critically on hardware parameters that these languages do not expose [3]. This is a shame. We want more algo-

rithms researchers to work on parallel algorithms, and they need better languages to do their work.

In the GPU niche of data-parallel languages, Obsidian is an exception [14], allowing for playfulness and invention on the low-level where you have (almost) complete control over the GPU, and still allowing computations to be composed efficiently using so called *pull*- and *push*-arrays. These arrays are not directly stored in a region of memory, but are rather representations of *array-computations*. This means that most array operations are cheap: they do not incur the overhead of writing a modified array to memory, but modifies the underlying symbolic array-computation directly. Obsidian uses a multi-staged compilation approach, which allows users to use Haskell as a meta-language generating Obsidian expressions. This can for instance be used to generate all the statements of an unrolled loop, or to precompute certain values already at code-generation time.

We present FCL, a reimplement of Obsidian with an external syntax implemented in Haskell2010 as a self-contained compiler¹. With FCL, we aim to allow the user to experiment and develop GPU algorithms in a composable fashion, with almost full control over the GPU hardware, for instance by allowing control of shared memory, distribution of work over blocks, warps and threads, memory coalescing, and kernel-fusion, which are main ingredients in many GPU algorithms.

In both Obsidian and FCL, computations are polymorphic in their mapping to executions on the GPU hardware, by the use of *level*-annotations in array types. We have developed a dynamic operational semantics for FCL that details the computational model and makes it clear how the different *levels* map to various iteration schemes on the GPU.

The rest of the paper is structured as follows. Section 2 explains *pull*- and *push*-arrays. In Section 3, we introduce FCL through three example programs: array reversal, matrix transpose, and parallel reduction. Section 4 we demonstrate that FCL is able to generate efficient OpenCL-code. Section 5 we do a rigorous introduction to FCL, defining its type system and dynamic semantics. Finally, we conclude in Section 7.

We did not find space for an introduction to GPU programming, we refer the reader to the OpenCL and CUDA programming guides by AMD [1] and NVIDIA [12].

[Copyright notice will appear here once 'preprint' option is removed.]

¹ FCL is available at <http://github.com/dybber/fcl>

2. Pull- and push-arrays

FCL inherits pull- and push-arrays from Obsidian [7]. As mentioned in the introduction, these are not actual arrays manifested in memory, but rather descriptions of how to produce an array. When the result of a pull- or push-array computation is written to memory, we say that the array has been *materialized*.

The two types of arrays complement each other: pull-arrays allows array indexing, but array concatenation and append are very inefficient. Push-arrays on the other hand allow for very efficient concatenation, but disallows array indexing.

The array representation in Obsidian is functional. Below we will introduce a simplified view of array representation in Obsidian, using Haskell notation.

2.1 Pull arrays

Let `Idx` be the type of array indices and array lengths. A pull-array with elements of type `a` is then represented as a length paired with an index:

```
type Pull a = (Idx, Idx -> a)
```

Materializing such an array in memory is performed by evaluating the function at each index and generating the code associated with writing the result to memory

Operating on the individual elements of the array can be done without materializing the array. Let `arr` be a function of the above type, multiplying each array element by two can then be done by building a new pull-array around it: `\i -> 2 * (arr i)`.

2.2 Push arrays

Push-arrays, on the other hand, already carry with them an *iteration pattern*, or *iteration scheme*, decided by the creator of that push-array. A push-array is represented by a function that can construct an array, when given a so-called *writer*-function. A *writer*-function is a function that accepts an element and an index and produces an assignment statement writing the element to its corresponding index in memory.

```
type Writer a = a -> Idx -> Program Thread ()
```

Here `Program Thread ()` is a computation in the used code-generation monad. Push-arrays are represented by a length and a function accepting such a writer:

```
type Push a t = (Idx, Writer a -> Program t ())
```

Materializing a push array is done by applying the function to a writer function, and the writer will then be invoked for each array element. This means that we can not access any single element of a push array, before it has been fully materialized.

In Obsidian iteration schemes on push-arrays are annotated in the array types, by a *level*-parameter, this is the `t` in the code above. The level-parameter can be either `grid`, `block`, `warp` or `thread`, corresponding to the hierarchy of organization for GPU threads, and annotates the sequential/parallel structure of the underlying iteration scheme. How levels are used will be explained in the context of FCL in the next section.

The main advantage of push-arrays compared with pull-arrays, is that they allow for efficient array appends. Where the append of two pull arrays will involve conditionals, the append of two push-arrays can be achieved by generating two separate loop structures and offsetting the writer function.

In FCL we keep the concepts of pull and push arrays, but abstract away from their actual representation, as will be illustrated in the rest of the paper.

3. Case studies in FCL

In this section we will demonstrate the use of FCL by implementing three different GPU algorithms, array reversal, array transposition using shared memory and parallel reduction.

3.1 Array reversal

Consider a program that reverses an array:

```
sig reverse : [a] -> [a]
fun reverse arr =
  let n = length arr
  in generate n (fn i => index arr (n - i - 1))
```

This program is implemented using the function `generate`, a language primitive that creates a new array by mapping the given function over the index-space $[0; n - 1]$. The program here *cannot* be compiled directly to GPU code, as it does not mention how it should be mapped to sequential or parallel loops. The arrays in this example are *pull*-arrays, and are identified by types of the form `[a]`, where `a` is a type variable, representing an arbitrary non-function type. To compile an FCL program into a kernel, we require the user to add an iteration scheme, detailing how this kernel should be mapped to the threads of the GPU. Such iteration schemes are annotated by a *level*, which can be either *thread* (sequential execution), *warp*, *block*, or *grid*. The iteration scheme is added using a function called `push`. Let us demonstrate, and create a block-level version of `reverse`.

```
sig revBlock : [a] -> [a]<block>
fun revBlock arr = push <block> (reverse arr)
```

Notice how the iteration scheme is reflected in the array type, `[a]<block>`. This is a *push*-array (from Obsidian). If we were to compile this function, FCL would generate a kernel reversing the entire array using a block-level computation. That is, the computation would only run in a single block, and thus only run on a single of the GPUs streaming multiprocessors. To distribute across several blocks, the input-arrays have to be partitioned and the resulting reversed array-chunks need to be concatenated back together again in the right order. In this case, the order of the chunks also needs to be reversed before concatenation.

```
sig revDistribute : int -> [a] -> [a]<grid>
fun revDistribute chunkSize arr =
  splitUp chunkSize arr
  |> map reverseBlock
  |> reverse
  |> concat chunkSize
```

The operator `|>` is reversed function application from F# and Elm, also known as forward-pipe. Notice that the same `reverse`-function can be used both to reverse the order of elements and the order of the blocks. The operation `concat` is what distributes the computation across a grid of blocks, as is also evident from its type:

```
concat : int -> [[a]<level>] -> [a]<1+level>
```

This means that each subarray is executed in a separate block, and `concat` makes sure that each block writes its result to adjacent subsections of the array it returns. Alternatively we could have applied `push <grid>` directly to the primitive `reverse` function, to add a grid-level iteration scheme to the array, but that is only possible in simple cases, where we do not need to manipulate the amount of data processed by each block or how results are combined. Neither `splitUp` nor `concat` is a primitive of FCL, and more complicated tiling and interleaving can thus be implemented, as we will see in the following example.




Figure 1: Transpose in Shared memory. Figure by NVIDIA.

3.2 Transpose in shared memory

Now consider the problem of matrix transposition. In FCL we only have one-dimensional arrays, which means that a two-dimensional matrix must be represented as its flat representation together with number of columns and rows. We are considering adding support for multidimensional-arrays, see Section 7.

If we follow a naive approach we can transpose a two-dimensional matrix, using the following `transpose`-function:

```
sig transpose : int -> int -> [a] -> [a]
fun transpose cols rows arr =
  generate (cols * rows)
    (fn n =>
      let i = n div rows
          j = n mod rows
      in index arr (j * rows + i))
```

If this version of `transpose` were to be executed in parallel on the GPU, it would lead to uncoalesced writes. When adding an iteration scheme to the final array, the final writes will always be in coalesced, but the indexing into the input array will not.

A more efficient approach is to chunk up the matrix in smaller 2D tiles, transpose each tile in shared memory, before stitching the tiles back together again (in transposed order). This approach makes both reads and writes to global memory coalesced, as the threads can first collaborate on moving data to shared memory, and afterwards collaborate on copying data from shared memory to the output-array.

The important thing to note is that this reading/writing order is encapsulated in `split2Dgrid` and `concat2Dgrid`, and a library of such operations can be provided to users.

```
sig transposeTiled : int -> int -> int ->
  [a] -> [a]<grid>
fun transposeTiled tileDim cols rows mat =
  let n = cols / tileDim
      m = rows / tileDim
  in split2DGrid tileDim cols n m mat
    |> map (force . push <block>)
    |> map (transpose tileDim tileDim)
    |> transpose n m
    |> map (push <block>)
    |> concat2DGrid tileDim n rows
```

This algorithm follows roughly the same structure as the `reverse`-example. However, instead of splitting the linear input-array into chunks (one following the other), we split and concatenate 2D tiles with the functions: `split2DGrid` and `concat2DGrid`. Also, we apply the function `force` which *executes* an iteration scheme, writing the array to shared memory, after which the array can again be indexed arbitrarily:

```
force : [a]<lvl> -> [a]
```

The result is a single kernel performing the transposition with all steps fused, performing just as well as the standard OpenCL implementation. For the sake of simplicity, the kernel in the form

presented here, works only for matrices that can be evenly divided by `tileDim`. To use this method for other matrices, a reshape operation increasing it's size can be performed and, the surplus columns and rows, can afterwards be removed using `drop`. We expect that these operations will be able to fuse, such that no additional reads/writes are necessary.

3.3 Parallel reduction

To implement a reduction kernel, we will perform a tree-reduction inside each work-group; this is implemented by splitting the subarray in two, and performing an element-wise sum of the two halves. This is very similar to what has previously been shown in Obsidian.

The FCL prelude provides the following functions for splitting arrays in two and joining arrays element-wise. These are not FCL primitives, but their implementation is standard and left out because of lack of space.

```
halve : [a] -> ([a], [a])
zipWith : (a -> b -> c) -> [a] -> [b] -> [c]
```

The tuple returned by `halve` is merely a syntactic construction. They will not be present in the OpenCL kernel code. Using these we can now write a function for taking one reduction-step:

```
sig step : <lvl> -> (a -> a -> a) ->
  [a] -> [a]<lvl>
fun step <lvl> f arr =
  let x = halve arr
  in push <lvl> (zipWith f (fst x) (snd x))
```

Notice that the function is polymorphic in the level-variable `lvl`. This makes it possible to postpone the decision of whether `step` will run sequentially or at one of the parallel levels of the hierarchy.

In Obsidian, we would have implemented this as a recursive function on the meta-level. As the function is doing a reduction on just a chunk of the array, we would statically know the size of this chunk, and Obsidian would generate an unrolled loop.

In FCL we instead provide a built-in looping-construct, `while`, which accepts a *stop-condition* and *stepping* function as arguments as well as the initial array.

```
sig red : <lvl> -> (a -> a -> a) -> [a] -> [a]<lvl>
fun red f arr =
  while (fn arr => 1 != lengthPull arr)
    (step <lvl> f)
    (step <lvl> f arr)
  |> push <lvl>
```

This will generate a while-loop, and automatically force values to shared memory between operations as well as performing a block-level synchronization between threads. In cases where the chunk size is known at compile time, we can use loop unrolling techniques to achieve the same code as if we had used Obsidian.

The `while`-construct assumes that arrays never need to grow during evaluation and thus reuses the same area of shared memory on each iteration. Also, `while` will always write the input array to shared memory before starting the iteration. To avoid doing a direct copy from global memory to shared memory, we take one initial step before starting the while-loop. This optimisation is called "First add during load" by Mark Harris [9].

To get this to run over multiple blocks, we need to split a larger array and concatenate the results:

```
sig reduceGrid : (a -> a -> a) -> [a] -> [a]<grid>
fun reduceGrid f arr =
  let chunkSize = 2 * #BlockSize
  in splitUp chunkSize arr
    |> map (red <block> f)
    |> concat 1
```




Figure 2: Measured bandwidths on our three example programs. OpenCL bars are code from NVIDIA’s OpenCL SDK, and we compare it to OpenCL kernels generated by FCL. The dashed line indicates the maximum bandwidth as measured by NVIDIA’s benchmarking tool.

Here `#BlockSize` will refer to either `get_local_size(0) / blockDim.x` or a constant specified by the user as configuration option at compilation time.

Another difference from Obsidian also comes to light here; as we no longer distinguish between statically known values and dynamically known values, we are not able to infer that `red <block> f` always returns a single scalar. We solve this by requiring an extra argument to `concat`, an expression computing the size of each chunk to concatenate.

4. Performance

FCL is work in progress; thus certain optimizations are still not implemented. However, the performance on the previously shown examples is promising, and we have identified the bottlenecks that are currently limiting performance.

To benchmark the generated code, we have used an NVIDIA GeForce GTX 780 Ti, which is built on the Kepler architecture. It has provides 2880 cores (875 Mhz), and 3GB GDDR5 ram (7 Ghz, bus-width: 384 bit). Calculating the theoretical peak bandwidth we get $7\text{Ghz} \times 384\text{bit} = 336\text{GB/s}$. In practice we can expect a 254.90GB/s maximum bandwidth, which we have measured using NVIDIA’s benchmarking tool (`bandwidthTest`).

Each benchmark has been executed on an array of 2^{24} 32-bit integers (67 MB). Timing was measured as wall-clock time on 1000 executions of the same kernel, preceded by a single warm-up run. The measured bandwidths are shown in Figure 2.

In the simple reverse example, we hit the measured maximum bandwidth as we hoped. The generated code is similar to the handwritten code by NVIDIA, except for block-virtualization, which is not used in NVIDIA’s version.

In the transpose example we are not quite on par with the handwritten code, and there are two reasons for that. First, we do not take care to avoid bank-conflicts, which we leave as future work. Second, we have quite a lot of extraneous divisions in the generated code. This is because we do not keep track of array shapes, and thus `split2DGrid` and `concat2DGrid` are performing some of the same work more than once. If we remove these double computations by hand, we achieve a performance boost, which is illustrated as FCL+handopt in the barplot. We are considering adding support for multi-dimensional arrays to tackle this issue, but this is also left as future work.

The reduction example is interesting; here we generate a completely unrolled loop, which performs reasonably well. To improve and reach the performance target set by NVIDIA’s heavily tuned kernel, we need to make each thread do an initial sequential reduction on a few elements, before the parallel tree-reduction we already have implemented.

5. Type system and semantics

To better understand the limitations, performance behavior and validate correctness of programs written in FCL, we will now turn to a more formal treatment of the language.

Previously, we have described both of the functions `concat` and `concat2DGrid`, which are used for distributing a computation. Both functions are written in terms of a more general operation, which we have named `interleave`, which in essence is a forward permutation on the indexes written to. However, in the limited treatment in this paper, we will focus on a simplified version of FCL with `concat` as a primitive, leaving out `concat2DGrid` and `interleave`. In all other aspects, this is a full treatment of FCL in its current state.

We use i, d and b to range over *integers*, *doubles*, and *booleans*, respectively. Let α range over an infinite set of type-variables and let x range over program variables. We use *unaryop* and *binaryop* to denote the sets of built in scalar operations.

Whenever z is some object, we write \vec{z} to range over sequences of similar objects. When we want to be explicit about the size of a sequence $\vec{z} = z_0, \dots, z_{(n-1)}$, we often write it on the form $\vec{z}^{(n)}$.

The core syntax of FCL is defined as follows:

```

op ::= unaryop | binaryop                (built-in operators)
    | generate | lengthPull | lengthPush
    | index | mapPush | mapPull
    | push | force | concat | while

bv ::= i | d | b                          (scalars)

γ ::= α | Z | 1 + γ                       (levels)

e ::= bv | x | [e1, ..., en] | op          (expressions)
    | fn x => e | fn ⟨α⟩ => e
    | e1 e2 | e ⟨γ⟩
    | let x = e1 in e2
    | (e1, e2) | fst e | snd e

```

Notice that the language has two application forms and two abstraction-forms; in addition to standard function application, we also have *level-application*, $e \langle \gamma \rangle$, for functions that accept a level-parameter. We often use the following short-hands for the first four levels:

```

thread = Z          block = 1 + (1 + Z)
warp    = 1 + Z      grid  = 1 + (1 + (1 + Z))

```

5.1 Type system

The syntax of FCL types, kinds and type-schemes is defined as follows:

```

bt ::= α | int | double | bool           (base types)
τ ::= α | bt | (τ1, τ2) | τ1 → τ2      (types)
    | ⟨α⟩ → τ
    | [τ]                                (pull arrays)
    | [bt]⟨γ⟩                            (push arrays)

κ ::= BT | GT | TYP | LVL               (kinds)
σ ::= forall α : κ. σ | τ                (type-schemes)

```

$\text{lengthPull} : [\alpha] \rightarrow \text{int}$
 $\text{lengthPush} : [\alpha]\langle \text{lvl} \rangle \rightarrow \text{int}$
 $\text{mapPull} : (\alpha \rightarrow \beta) \rightarrow [\alpha] \rightarrow [\beta]$
 $\text{mapPush} : (\alpha \rightarrow \beta) \rightarrow [\alpha]\langle \text{lvl} \rangle \rightarrow [\beta]\langle \text{lvl} \rangle$
 $\text{generate} : \text{int} \rightarrow (\text{int} \rightarrow \alpha) \rightarrow [\alpha]$
 $\text{index} : [\alpha] \rightarrow \text{int} \rightarrow \alpha$
 $\text{push} : \langle \text{lvl} \rangle \rightarrow [\alpha] \rightarrow [\alpha]\langle \text{lvl} \rangle$
 $\text{force} : [\alpha]\langle \text{lvl} \rangle \rightarrow [\alpha]$
 $\text{while} : ([\alpha] \rightarrow \text{bool}) \rightarrow ([\alpha] \rightarrow [\alpha]\langle \text{lvl} \rangle) \rightarrow [\alpha]\langle \text{lvl} \rangle \rightarrow [\alpha]$

Figure 3: Types of built-in operators.

The types of built-in array combinators are shown in Figure 3.

To define the set of valid types, we define a relation $\Delta \vdash \tau$ below, where Δ are kind environments, mapping type variables to kinds:

$$\Delta ::= \alpha : \kappa, \Delta \mid \epsilon$$

The kind-system divides types into four categories. Base types (BT), ground types (GT), general types (TYP) and levels (LVL). Basetypes are types of scalar values, which are the only types of values allowed in push-arrays. Ground types are all types except function-types, and are the types allowed in pull-arrays.

Kind system

$$\Delta \vdash \tau : \kappa$$

$$\begin{aligned}
& \Delta \vdash \text{int} : \text{BT} \quad (1) \quad \Delta \vdash \text{double} : \text{BT} \quad (2) \quad \Delta \vdash \text{bool} : \text{BT} \quad (3) \\
& \frac{\Delta(\alpha) = \kappa}{\Delta \vdash \alpha : \kappa} \quad (4) \quad \frac{\Delta \vdash \tau_i : \kappa \quad \kappa \neq \text{LVL}}{\Delta \vdash (\tau_1, \tau_2) : \kappa} \quad (5) \\
& \frac{\Delta \vdash \tau : \text{BT}}{\Delta \vdash \tau : \text{GT}} \quad (6) \quad \frac{\Delta \vdash \tau : \text{GT}}{\Delta \vdash \tau : \text{TYP}} \quad (7) \\
& \frac{}{\Delta \vdash \tau \rightarrow \tau' : \text{TYP}} \quad (8) \quad \frac{\Delta, \alpha : \text{LVL} \vdash \tau : \text{TYP}}{\Delta \vdash \langle \alpha \rangle \rightarrow \tau : \text{TYP}} \quad (9) \\
& \frac{\Delta \vdash \tau : \text{GT}}{\Delta \vdash [\tau] : \text{GT}} \quad (10) \quad \frac{\Delta \vdash \tau : \text{BT}}{\Delta \vdash [\tau]\langle \gamma \rangle : \text{GT}} \quad (11) \\
& \frac{}{\Delta \vdash Z : \text{LVL}} \quad (12) \quad \frac{\Delta \vdash \gamma : \text{LVL}}{\Delta \vdash 1 + \gamma : \text{LVL}} \quad (13)
\end{aligned}$$

A type environment Γ , is a set of type assumptions of the form $x : \sigma$, mapping program variables to type-schemes:

$$\Gamma ::= x : \sigma, \Gamma \mid \epsilon$$

We define the relation $\sigma \succ_{\Delta} \sigma'$ to denote that a type scheme σ' is an instance of another type scheme σ .

$$\begin{aligned}
& \frac{\Delta \vdash \tau : \kappa \quad \kappa \neq \text{LVL}}{\forall \alpha. \sigma \succ_{\Delta} \sigma[\alpha \mapsto \tau]} \quad (14) \\
& \frac{}{\sigma \succ_{\Delta} \sigma} \quad (15) \quad \frac{\sigma \succ_{\Delta} \sigma' \quad \sigma' \succ_{\Delta} \sigma''}{\sigma \succ_{\Delta} \sigma''} \quad (16)
\end{aligned}$$

The type system allows inferences among sentences of the form $\Delta, \Gamma \vdash_{\gamma} e : \tau$, which are read: “under the assumptions Δ, Γ the expression e has type τ at level γ ”. The typing rules are shown on the opposing page. The γ annotation on the turnstyle, is used to restrict how array computations can be nested. In all other rules than the rule for `concat`, γ is passed on unchanged, but in expression below a `concat` only operations on a lower level can be used.

Expression typing

$$\Delta, \Gamma \vdash_{\gamma} e : \tau$$

$$\frac{}{\Delta, \Gamma \vdash_{\gamma} i : \text{int}} \quad (17) \quad \frac{}{\Delta, \Gamma \vdash_{\gamma} d : \text{double}} \quad (18)$$

$$\frac{}{\Delta, \Gamma \vdash_{\gamma} b : \text{bool}} \quad (19)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_i : \tau, \text{ for all } i \quad \Delta \vdash_{\gamma} \tau : \text{GT}}{\Delta, \Gamma \vdash_{\gamma} [e_1, \dots, e_n] : [\tau]} \quad (20)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_1 : \tau_1 \quad \Delta, \Gamma \vdash_{\gamma} e_2 : \tau_2}{\Delta, \Gamma \vdash_{\gamma} (e_1, e_2) : (\tau_1, \tau_2)} \quad (21)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e : (\tau_1, \tau_2)}{\Delta, \Gamma \vdash_{\gamma} \text{fst } e : \tau_1} \quad (22) \quad \frac{\Delta, \Gamma \vdash_{\gamma} e : (\tau_1, \tau_2)}{\Delta, \Gamma \vdash_{\gamma} \text{snd } e : \tau_2} \quad (23)$$

$$\frac{\Gamma(x) = \sigma \quad \sigma \succ_{\Delta} \tau \quad \Delta \vdash_{\gamma} \tau : \text{TYP}}{\Delta, \Gamma \vdash_{\gamma} x : \tau} \quad (24)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_1 : \tau \quad \vec{\alpha} = \text{ftv}(\tau) \quad \Delta, (\Gamma, x : \forall \vec{\alpha}. \tau) \vdash_{\gamma} e_2 : \tau}{\Delta, \Gamma \vdash_{\gamma} \text{let } x = e_1 \text{ in } e_2 : \tau} \quad (25)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_1 : \tau' \rightarrow \tau \quad \Delta, \Gamma \vdash_{\gamma} e_2 : \tau'}{\Delta, \Gamma \vdash_{\gamma} e_1 e_2 : \tau} \quad (26)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e : \langle \alpha \rangle \rightarrow \tau \quad \Delta \vdash_{\gamma} \alpha : \text{LVL}}{\Delta, \Gamma \vdash_{\gamma} e \langle \gamma \rangle : \tau[\alpha \mapsto \gamma]} \quad (27)$$

$$\frac{\Delta, (\Gamma, x : \tau') \vdash_{\gamma} e : \tau}{\Delta, \Gamma \vdash_{\gamma} \text{fn } x \Rightarrow e : \tau' \rightarrow \tau} \quad (28)$$

$$\frac{(\Delta, \alpha : \text{LVL}), \Gamma \vdash_{\gamma} e : \tau}{\Delta, \Gamma \vdash_{\gamma} \text{fn } \langle \alpha \rangle \Rightarrow e : \langle \alpha \rangle \rightarrow \tau} \quad (29)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_1 : \text{int} \quad \Delta, \Gamma \vdash_{\gamma} e_2 : [[\alpha]\langle \gamma \rangle]}{\Delta, \Gamma \vdash_{1+\gamma} \text{concat } e_1 e_2 : [\alpha]\langle 1 + \gamma \rangle} \quad (30)$$

5.2 Dynamic semantics

We now turn towards the semantics of the language, which will help understand compilation of FCL terms and how the level-types instruct the compilation.

The evaluation relation we will define below, is annotated with a location. Locations emulate the hierarchical structure of a parallel machine, and are of the form:

$$\begin{aligned}
\text{loc} ::= & \text{Thread}(\text{thread_id}) \quad \text{thread_id} \in \mathbb{N} \\
& \mid \text{Group}(\{\text{loc}_1, \dots, \text{loc}_n\})
\end{aligned}$$

Locations relates to levels, and we introduce similar shorthands.

$$\text{Warp}(\vec{\text{loc}}) = \text{Group}(\{\text{Thread}(\text{loc}_1), \dots, \text{Thread}(\text{loc}_n)\})$$

$$\text{Block}(\vec{\text{loc}}) = \text{Group}(\{\text{Warp}(\text{loc}_1), \dots, \text{Warp}(\text{loc}_n)\})$$

$$\text{Grid}(\vec{\text{loc}}) = \text{Group}(\{\text{Block}(\text{loc}_1), \dots, \text{Block}(\text{loc}_n)\})$$

We also define whether a location is *respecting* a level, by the relation, $\text{loc} \triangleright \gamma$, defined by the following two rules:

$$\frac{}{\text{Thread}(\text{thread_id}) \triangleright Z} \quad (31) \quad \frac{\text{loc}_i \triangleright \gamma \text{ for all } i}{\text{Group}(\text{loc}) \triangleright 1 + \gamma} \quad (32)$$

Values in FCL are either base values (*bv*), pull arrays, push arrays or delayed concatenation of push-arrays.

$$\begin{aligned}
v ::= & \text{bv} && \text{(base values)} \\
& \mid [e_1, \dots, e_n] && \text{(pull array)} \\
& \mid [e_1, \dots, e_n]\langle \gamma \rangle && \text{(push array)} \\
& \mid \text{concatDelay } e_1 e_2 && \text{(delayed concat)}
\end{aligned}$$

We extend the typing relation above to include typing of values.

Value typing

$$\boxed{\Delta, \Gamma \vdash_{\gamma} v : \tau}$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_i : \tau \quad \Delta \vdash_{\gamma} \tau : \mathbf{GT}}{\Delta, \Gamma \vdash_{\gamma} [e_1, \dots, e_n] : [\tau]} \quad (33)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_i : \tau \quad \Delta \vdash_{\gamma} \tau : \mathbf{BT}}{\Delta, \Gamma \vdash_{\gamma} [e_1, \dots, e_n] \langle \gamma \rangle : [\tau] \langle \gamma \rangle} \quad (34)$$

$$\frac{\Delta, \Gamma \vdash_{\gamma} e_1 : \mathbf{int} \quad \Delta, \Gamma \vdash_{\gamma} e_2 : [\tau] \langle \gamma \rangle}{\Delta, \Gamma \vdash_{1+\gamma} \mathbf{concatDelay} e_1 e_2 : [\tau] \langle 1 + \gamma \rangle} \quad (35)$$

We now define the promised dynamic semantics of FCL. Due to space limitations, we consider just the interesting cases involving `force`. The first two rules are administrative fusion rules, happening at compile time, there are a bunch more of these, which we are not able to display.

Small-step semantics

$$\boxed{e \hookrightarrow_{loc} e'}$$

$$\frac{}{\mathbf{mapPull} e [e_1, e_2, \dots, e_n] \hookrightarrow_{loc} [e e_1, e e_2, \dots, e e_n]} \quad (36)$$

$$\frac{}{\mathbf{mapPush} e [e_1, e_2, \dots, e_n] \langle \gamma \rangle \hookrightarrow_{loc} [e e_1, e e_2, \dots, e e_n] \langle \gamma \rangle} \quad (37)$$

$$\frac{}{\mathbf{concat} e_1 e_2 \hookrightarrow_{loc} \mathbf{concatDelay} e_1 e_2} \quad (38)$$

$$\frac{}{\mathbf{force} [bv_1, \dots, bv_n] \langle lvl \rangle \hookrightarrow_{loc} [bv_1, \dots, bv_n]} \quad (39)$$

$$\frac{e_i \hookrightarrow_{\mathbf{Thread}(t)} e'_i}{\mathbf{force} [bv_1, \dots, e_i, \dots, e_n] \langle \mathbf{thread} \rangle \hookrightarrow_{\mathbf{Thread}(t)} \mathbf{force} [bv_1, \dots, e'_i, \dots, e_n] \langle \mathbf{thread} \rangle} \quad (40)$$

$$\frac{e_i \hookrightarrow_{loc_i} e'_i \text{ for all } i \in [1, n]}{\mathbf{force} [e_1, \dots, e_n] \langle \mathbf{lvl} \rangle \hookrightarrow_{\mathbf{Group}(loc)} \mathbf{force} [e'_1, \dots, e'_n] \langle \mathbf{lvl} \rangle} \quad (41)$$

$$\frac{e \hookrightarrow_{loc} m \quad \mathbf{force} [\vec{e}_i] \langle lvl \rangle \hookrightarrow_{loc_i}^* [\vec{bv}_i] \text{ for all } i \in [1, n]}{\mathbf{force} (\mathbf{concatDelay} e [[\vec{e}_1], [\vec{e}_2], \dots, [\vec{e}_n]] \langle lvl \rangle) \hookrightarrow_{\mathbf{Group}(loc)} [\vec{bv}_1, \vec{bv}_2, \dots, \vec{bv}_n]} \quad (42)$$

Only programs of type $[\alpha] \langle \gamma \rangle$ can be fully evaluated under these semantics. For instance, we will require a reduction-kernel will return a singleton array instead of an integer. This is by intention, we want all programs to have explicit hierarchy-annotations, describing the level of execution, and currently this is only allowed for push-arrays.

PROPOSITION 1 (Type Preservation). *If $\Delta, \Gamma \vdash_{\gamma} e : \tau$ and $e \hookrightarrow_{loc} e'$ for some location $loc \triangleright \gamma$, then $\Delta, \Gamma \vdash_{\gamma} e' : \tau$.*

PROPOSITION 2 (Progress). *If $\Delta, \Gamma \vdash_{\gamma} e : \tau$, then either e is a value or $e \hookrightarrow_{loc} e'$ for some e' .*

6. Related work

FCL builds on previous work on Obsidian [14], from which both the concepts of push-arrays and level-variables originates. The language discussed by Dubach et al. [13] is also related, operating at a similarly low-level. It might be interesting to build a similar set of rewrite rules on top of FCL, and search for good rewrites.

The hierarchy in FCL and Obsidian, might also be compared to the concept of locales and sublocales in the Chapel language [6].

As previously mentioned other work has been done in the area of functional languages for GPU computing, but most efforts have been on optimizing compilers. This includes Futhark [10], Accelerate [5], Delite [4], Harlan [11], and Nessie [2].

7. Conclusion and Future work

We have presented FCL, a functional language for GPU algorithms. FCL is work in progress. Currently only device-code is generated, and host-code have to be written manually, but we plan to get that supported later this year. In addition, memory is currently allocated implicitly, and it thus not possible to reuse the same memory. We would want the possibility of writing an in-place version of `reverse`, writing the reserved array back to the same global-array.

Our limitation of only having one-dimensional arrays, will in many cases lead to unnecessary shape-computations, as we saw in the transpose example. We will thus investigate how shapes can be introduced, such that `split2DGrid`, would split the array into a 2D array of 2D arrays.

Future work also includes bank-conflict avoidance, use of vectorised GPU-instructions, and the addition of sequential loops with array updates, perhaps in the style of Futhark [10].

Finally, we would like to implement some larger example programs in FCL, and attempt using FCL as an intermediate language for our APL-compiler [8].

References

- [1] AMD Accelerated Parallel Processing, *OpenCL Programming Guide*. Advanced Micro Devices, Inc., 2013.
- [2] L. Bergstrom and J. Reppy. Nested Data-parallelism on the GPU. In *Proceedings of the 17th ACM SIGPLAN International Conference on Functional Programming, ICFP '12*. ACM, 2012.
- [3] P. Carlsen and M. Dybdal. Option pricing using data-parallel languages. Master's thesis, DIKU, University of Copenhagen, Department of Computer Science, 2013.
- [4] H. Chafi, A. K. Sujeeth, K. J. Brown, H. Lee, A. R. Atreya, and K. Olukotun. A domain-specific approach to heterogeneous parallelism. In *ACM SIGPLAN Notices*, volume 46. ACM, 2011.
- [5] M. M. Chakravarty, G. Keller, S. Lee, T. L. McDonell, and V. Grover. Accelerating Haskell array codes with multicore GPUs. In *6th Workshop on Decl. Aspects of Multicore Programming, DAMP'11*. ACM, 2011.
- [6] B. L. Chamberlain, D. Callahan, and H. P. Zima. Parallel programmability and the Chapel language. *International Journal of High Performance Computing Applications*, 21(3):291–312, 2007.
- [7] K. Claessen, M. Sheeran, and B. J. Svensson. Expressive array constructs in an embedded GPU kernel programming language. In *Proceedings of the 7th workshop on Declarative aspects and applications of multicore programming*, pages 21–30. ACM, 2012.
- [8] M. Elsmann and M. Dybdal. Compiling a subset of APL into a typed intermediate language. In *1st Int. Workshop on Libraries, Languages and Compilers for Array Programming, ARRAY'14*. ACM, 2014.
- [9] M. Harris et al. Optimizing parallel reduction in cuda. *NVIDIA Developer Technology*, 2(4), 2007.
- [10] T. Henriksen and C. E. Oancea. A T2 graph-reduction approach to fusion. In *Proceedings of the 2nd ACM SIGPLAN workshop on Functional high-performance computing*, pages 47–58. ACM, 2013.
- [11] E. Holk, W. E. Byrd, N. Mahajan, J. Willcock, A. Chauhan, and A. Lumsdaine. Declarative Parallel Programming for GPUs. In *International Conference on Parallel Computing (ParCo 2011)*, 2011.
- [12] NVIDIA. CUDA C Programming Guide, 2015.
- [13] M. Steuwer, C. Fensch, S. Lindley, and C. Dubach. Generating performance portable code using rewrite rules: From high-level functional expressions to high-performance opencl code. In *Proceedings of the 20th ACM SIGPLAN International Conference on Functional Programming, ICFP 2015*, 2015.
- [14] B. J. Svensson, R. R. Newton, and M. Sheeran. A language for hierarchical data parallel design-space exploration on GPUs. *Journal of Functional Programming*, 26, 2016.